

Subword complexity of the Fibonacci-Thue-Morse sequence: the proof of Dekking's conjecture

Jeffrey Shallit*

School of Computer Science
University of Waterloo
Waterloo, ON N2L 3G1
Canada
shallit@uwaterloo.ca

October 20 2020

Abstract

Recently Dekking conjectured the form of the subword complexity function for the Fibonacci-Thue-Morse sequence. In this note we prove his conjecture by purely computational means, using the free software *Walnut*.

1 Introduction

Recall that the Fibonacci numbers F_n are defined by $F_0 = 0$, $F_1 = 1$, and $F_n = F_{n-1} + F_{n-2}$.

In this note we will use the Fibonacci (or “Zeckendorf”) representation of natural numbers [6, 9]. This is a map from \mathbb{N} to the set of binary strings, written $(n)_F$, such that

- (a) If $(n)_F = w$, and $t = |w|$, then $n = \sum_{1 \leq i \leq t} w[i] F_{t+2-i}$,
- (b) $w[1]$ (if it exists) is 1;
- (c) w does not have two consecutive 1's.

The inverse map is written $[w]_F$.

The so-called *Fibonacci-Thue-Morse* sequence $(\mathbf{ftm}[n])_{n \geq 0}$ is defined to be the sum, taken modulo 2, of the bits of the Fibonacci representation of n . See, for example, [1, Examples 7.8.2, 7.8.4] and [4]. This is sequence [A095076](#) in the *On-Line Encyclopedia of Integer Sequences* (OEIS). The first few terms are as follows:

*Supported by NSERC Grant 2018-04118.

| n | 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9 | 10 | 11 | 12 | 13 | 14 | 15 | 16 | 17 | 18 | 19 |
|-------------------|---|---|---|---|---|---|---|---|---|---|----|----|----|----|----|----|----|----|----|----|
| $\mathbf{ftm}[n]$ | 0 | 1 | 1 | 1 | 0 | 1 | 0 | 0 | 1 | 0 | 0 | 0 | 1 | 1 | 0 | 0 | 0 | 1 | 0 | 1 |

The sequence \mathbf{ftm} is *Fibonacci-automatic*. This means there is an automaton that, on input the Fibonacci representation of n , computes $\mathbf{ftm}[n]$. In fact, it is easy to see that this automaton has 4 states, as follows:




Figure 1: Fibonacci automaton for the sequence \mathbf{ftm}

Each state is labeled with the syntax “name/output”.

The *subword complexity* function $\rho_{\mathbf{s}}(n)$ of a sequence \mathbf{s} is defined to be the number of distinct length- n contiguous blocks appearing in \mathbf{s} . Recently Dekking [2] gave a conjecture on the form of $\rho_{\mathbf{ftm}}(n)$. In this note we prove this conjecture.

2 Dekking’s conjecture

One form of Dekking’s conjecture involves the first difference $d(n) := \rho_{\mathbf{ftm}}(n+1) - \rho_{\mathbf{ftm}}(n)$ of the subword complexity function. It is known that $d(n)$ counts the number of *right-special* factors, where a factor x is right-special if both $x0$ and $x1$ are factors of \mathbf{ftm} .

Conjecture 1 (Dekking).

$$\begin{aligned}
(d(n))_{n \geq 0} &= 1, 2, 4, 6, 10, 6, 6, 8, 6, 8, 8, 6, 6, 6, 6, 8, 8, 8, 6, 6, 6, \dots \\
&= 1, 2, 4, 6, 10, 6^2, 8^1, 6^1, 8^2, 6^4, 8^3, 6^4, 8^5, 6^9, 8^8, 6^{12}, 8^{13}, \dots \\
&= 1, 2, 4, 6, 10 \prod_{i \geq 2} 6^{F_i + (-1)^i} 8^{F_i}.
\end{aligned}$$

Here the exponents on the numbers denote repetition factors, and the \prod symbol denotes concatenation.

3 Outline of the proof

We can prove Dekking’s conjecture via purely computational means, using a decision procedure implemented in the free software **Walnut** [7].

There are five steps in the proof.

1. We create a first-order logical formula specifying that the length- n factor $\mathbf{ftm}[i..i+n-1]$ beginning at position i is a right-special factor.
2. Using the ideas in [8], we can create a deterministic finite automaton (DFA) accepting the Fibonacci representation of the pairs (i, n) such that
 - (a) $\mathbf{ftm}[i..i+n-1]$ is a right-special factor; and
 - (b) $\mathbf{ftm}[i..i+n-1]$ is the earliest occurrence of that factor in \mathbf{ftm} .
3. Using the ideas in [3], we can read off, from the automaton in the previous step, the *linear representation* for $d(n)$. This consists of vectors v, w and a matrix-valued morphism γ such that $d(n) = v\gamma((n)_F)w$ for all $n \geq 0$. The *rank* of the linear representation is the size of v .
4. Using the so-called “semigroup trick” [3], we can prove that the semigroup generated by the two matrices $\gamma(0), \gamma(1)$ is finite. Furthermore, we can create a Fibonacci-DFAO computing $d(n)$.
5. This DFAO can be minimized using the standard methods [5], and Dekking’s result can be read off the result.

4 Details of the proof

Now we provide a few more details about each step.

1. The formula is created in several steps. First, we need a formula asserting that $\mathbf{ftm}[i..i+n-1] = \mathbf{ftm}[j..j+n-1]$.

$$\mathbf{ftmfactoreq}(i, j, n) := \forall t (t < n) \implies \mathbf{ftm}[i+t] = \mathbf{ftm}[j+t].$$

Next, we create a formula asserting that $\mathbf{ftm}[i..i+n-1]$ is right-special:

$$\mathbf{isftmrs}(i, n) := \exists j, k \ \mathbf{ftmfactoreq}(i, j, n) \wedge \mathbf{ftmfactoreq}(i, k, n) \wedge \mathbf{ftm}[j+n] \neq \mathbf{ftm}[k+n].$$

Finally, we create a formula asserting that $\mathbf{ftm}[i..i+n-1]$ is right-special and also is the first occurrence of that factor:

$$\mathbf{ftmrsn}(i, n) := \mathbf{isftmrs}(i, n) \wedge \forall j (j < i) \implies \neg \mathbf{ftmfactoreq}(i, j, n).$$

2. We can translate the formulas above into automata using the **Walnut** system. The automata recognize the Fibonacci representations of the values of the free variables in the formulas that make the formula true. We do this with the following **Walnut** commands, which are more or less verbatim translations of the predicates above.

```

def ftmfactoreq "?msd_fib At (t<n) => FTM[i+t]=FTM[j+t]":
def isftmrs "?msd_fib Ej,k $ftmfactoreq(i,j,n) & $ftmfactoreq(i,k,n)
    & FTM[j+n] != FTM[k+n]":
def ftmrns "?msd_fib $isftmrs(i,n) & Aj (j<i) => ~$ftmfactoreq(i,j,n)":

```

Walnut needed 80 Gigs of memory to do this computation, and used 48 minutes of computation on a 2 CPU-Intel E5-2697 v3 Xeon machine with 256GB Ram. (To do this, start Walnut with

```
java -Xmx85000M Main.prover
```

on the command line.) This is one of the largest computations we have done with Walnut.

The DFA for `ftmfactoreq` has 91 states, the DFA for `isftmrs` has 45 states, and the DFA for `ftmrns` has 46 states.

3. We can create the linear representation with the Walnut command

```
eval ftmatrix n "?msd_fib $ftmrns(i,n)":
```

This gives us a linear representation (v, γ, w) of rank 46, which is stored in the file `ftmatrix.mpl`.

For technical reasons, this representation now needs to be adjusted by replacing v with $v\gamma(000)$.

4. The “semigroup trick” gives a DFAO with 75 states.
5. After minimization, the Fibonacci-DFAO produced is below.




Figure 2: Fibonacci DFAO computing $d(n)$

5 Deducing Dekking's result from the automaton

Theorem 2. *We claim that*

$$\begin{aligned} d(n) &= 8 \text{ for } F_{2m+1} + 2 \leq n \leq F_{2m+1} + F_{2m-2} + 1 \\ d(n) &= 6 \text{ for } F_{2m+1} + F_{2m-2} + 2 \leq n \leq F_{2m+2} \\ d(n) &= 8 \text{ for } F_{2m+2} + 1 \leq n \leq F_{2m+2} + F_{2m-1} \\ d(n) &= 6 \text{ for } F_{2m+2} + F_{2m-1} + 1 \leq n \leq F_{2m+3} + 1 \end{aligned}$$

for $m \geq 2$.

Proof. We can prove this with Walnut, too. The first step is to take the DFAO given in Figure 2, and add it to Walnut's Word Automata Library. We'll use the name FTMD for this DFAO. It is given in Appendix A.

Next, we will create individual formulas for each of the intervals given in the statement of the theorem; we'll call them `interval1`, ..., `interval4`. In order to specify them we need ways to specify that x is a Fibonacci number, an odd-indexed Fibonacci number, an even-indexed Fibonacci number, or that two numbers represent consecutive Fibonacci numbers. These are done, respectively, with the formulas `fibo`, `oddfib`, `evenfib`, and `consecfib`. We also need a formula specifying that x is of the form $F_n + F_{n-3}$; this is done with `fib1001`. We can verify that our intervals cover all integers $n \geq 7$ with the formula `all`.

Finally, we perform four tests, one for each interval, checking each of the assertions in the statement of the theorem. Each of these tests returns `true`, so the theorem is proven.

Here is the Walnut code:

```
reg fibo msd_fib "0*10*":
reg oddfib msd_fib "0*10(00)*":
reg evenfib msd_fib "0*1(00)*":
def consecfib "?msd_fib (t<u) & $fibo(t) & $fibo(u) & Av (t<v & v<u) => ~$fibo(v)":
reg fib1001 msd_fib "0*10010*":

def interval1 "?msd_fib Et,u,y $oddfib(t) & $evenfib(u) & $consecfib(t,u) &
  $fib1001(y) & (t<y) & (y<=u) & (t+2<=n) & (n<=y+1)":
def interval2 "?msd_fib Et,u,y $oddfib(t) & $evenfib(u) & $consecfib(t,u) &
  $fib1001(y) & (t<y) & (y<=u) & (y+2<=n) & (n<=u)":
def interval3 "?msd_fib Ew,x,z $evenfib(x) & $oddfib(z) & $consecfib(x,z) &
  $fib1001(w) & (x<w) & (w<=z) & (x+1<=n) & (n<=w)":
def interval4 "?msd_fib Ew,x,z $evenfib(x) & $oddfib(z) & $consecfib(x,z) &
  $fib1001(w) & (x<w) & (w<=z) & (w+1<=n) & (n<=z+1)":

def all "?msd_fib $interval1(n) | $interval2(n) | $interval3(n) | $interval4(n)":

eval test1 "?msd_fib An $interval1(n) => FTMD[n]=@8":
eval test2 "?msd_fib An $interval2(n) => FTMD[n]=@6":
```

```
eval test3 "?msd_fib An $interval3(n) => FTMD[n]=@8":
eval test4 "?msd_fib An $interval4(n) => FTMD[n]=@6":
```

□

Corollary 3. *Dekking's conjecture is correct.*

References

- [1] J.-P. Allouche and J. O. Shallit, *Automatic Sequences*, Cambridge University Press, 2003.
- [2] F. M. Dekking, The structure of Zeckendorf representations and base φ expansions, talk for the One World Seminar on Combinatorics on Words, October 19 2020. Available at http://www.i2m.univ-amu.fr/wiki/Combinatorics-on-Words-seminar/_media/seminar2020:slides20201019dekking.pdf.
- [3] C. F. Du, H. Mousavi, L. Schaeffer, and J. Shallit. Decision algorithms for Fibonacci-automatic words III: Enumeration and abelian properties. *Internat. J. Found. Comp. Sci.* **27** (2016), 943–963.
- [4] E. Ferrand, An analogue of the Thue-Morse sequence, *Electronic J. Combinatorics* **14** (2007), Paper R30. Available at <https://doi.org/10.37236/948>.
- [5] John E. Hopcroft and Jeffrey D. Ullman. *Introduction to Automata Theory, Languages, and Computation*. Addison-Wesley, 1979.
- [6] C. G. Lekkerkerker. Voorstelling van natuurlijke getallen door een som van getallen van Fibonacci. *Simon Stevin* **29** (1952), 190–195.
- [7] H. Mousavi. Automatic theorem proving in Walnut. Available at <http://arxiv.org/abs/1603.06017>, 2016.
- [8] H. Mousavi, L. Schaeffer, and J. Shallit. Decision algorithms for Fibonacci-automatic words, I: Basic results. *RAIRO Inform. Théor. App.* **50** (2016), 39–66.
- [9] E. Zeckendorf. Représentation des nombres naturels par une somme de nombres de Fibonacci ou de nombres de Lucas. *Bull. Soc. Roy. Liège* **41** (1972), 179–182.

Appendix A Walnut file FTMD.txt

msd_fib

0 1
0 -> 0
1 -> 1
1 2
0 -> 2
2 4
0 -> 3
1 -> 4
3 6
0 -> 5
1 -> 6
4 10
0 -> 7
5 6
0 -> 8
1 -> 9
6 6
0 -> 10
7 8
0 -> 11
1 -> 12
8 6
0 -> 5
1 -> 13
9 8
0 -> 14
10 8
0 -> 15
1 -> 16
11 6
0 -> 11
1 -> 12
12 6
0 -> 11
13 6
0 -> 14
14 8
0 -> 14
1 -> 9
15 8
0 -> 10
1 -> 12
16 8
0 -> 11