

Type-Directed Partial Evaluation in Haskell^{*}

Kristoffer.Rose@ENS-Lyon.fr[†]

June 10, 1998 (version 1.19)

Abstract

We implement simple standard *type-directed partial evaluation* in the pure functional programming language Haskell, using type classes.

1 Introduction

In this paper we implement *standard type-directed partial evaluation* (TDPE) of Danvy (1996) for a small subset of the pure (lazy) functional programming language Haskell (Fasel, Hudak, Peyton Jones, Wadler et al. 1992). When summarizing the definitions we rely on *combinatory reduction systems* (CRS) of Klop, van Oostrom & van Raamsdonk (1993), however, one can read the specification without knowledge of CRSs.

The document is itself a program written in “literate Haskell:” lines that begin with small numbers 1 2 ... are Haskell program lines, except we use the symbols \rightarrow , \Rightarrow , \equiv , \times , \uparrow , and λ instead of the usual Haskell equivalents \rightarrow , $=>$, $=$, $*$, $^$, and \backslash , and let $_$ denote a blank space inside a string.

The first line of Haskell identifies the code.

```
1 module TDPE where
```

In section 2 we introduce the base “2-level” calculus that constitutes our subset language and for which we do normalization. The following section 3 shows how we implement TDPE using Haskell type classes; section 4 shows how this can be used to implement usual partial evaluation. Finally, section 5 shows several small examples and section 6 concludes.

^{*}Presented at the “Normalisation by Evaluation 1998” APPSEM workshop, Chalmers, Göteborg.

[†]Address: LIP, École Normale Supérieure de Lyon, 46 Allée d’Italie, F-69364 Lyon 7, France.

Home page: <http://www.ens-lyon.fr/~krisrose>.

2 2-level λ -calculus

TDPE works with 2-level λ -calculus (Nielson & Nielson 1992) as a means to study the two “binding times” of partial evaluation. TDPE uses this same idea, however, instead of seeing the levels as data annotations we see them as marks distinguishing data and code.

2.1 Terms

The types and 2-level terms used for standard TDPE are defined as follows (using higher-order abstract syntax in CRS¹ notation):

$$\begin{aligned}
 \alpha, \beta &::= b \mid \alpha \rightarrow \beta \mid \alpha \times \beta && \text{(Type)} \\
 E, F &::= c \mid \underline{\lambda}x. E[x] \mid (F _ E) \mid \underline{\text{pair}}(E_1, E_2) \mid \underline{\text{fst}} E \mid \underline{\text{snd}} E && \text{(Expr)} \\
 V, W &::= c \mid \overline{\lambda}z. V[z] \mid (V \overline{_} W) \mid \overline{\text{pair}}(V_1, V_2) \mid \overline{\text{fst}} V \mid \overline{\text{snd}} V && \text{(Value)}
 \end{aligned}$$

where b denotes a base type and c is a constant of base type.

Note how the use of higher-order abstract syntax means that we do not need a case for variables: Their meaning is clear from the use of “ $E[x]$ ” that can be read as “inside the subterm E the variable x may occur as a subterm.”

When realising this in Haskell we represent types by themselves, *i.e.*, as the corresponding Haskell type, values are native Haskell values, and expressions are a data type using a higher-order function for the binding. This means that we cannot rely on the default Show instance to print expressions (the appendix has the one we use).

```

2 data Expr = LAMBDA (Expr → Expr)
3           | APPLY Expr Expr
4           | PAIR Expr Expr
5           | FST Expr
6           | SND Expr
7           | BASE String

```

The last is for free variables, constants, *etc.*: all the things where we don’t really care what they look like as long as they print as the right thing.

¹Except we use non-terminal letters as meta-variables rather than the ubiquitous Z , write abstractions as $_ _$, and meta-application as $_ _$.

2.2 Static Reductions

Values are reduced as follows (again in CRS notation):

$$\begin{aligned} (\overline{\lambda x}. V[x]) \overline{W} &\rightarrow V[W] & (\overline{\beta}) \\ \overline{\text{fst}} (\overline{\text{pair}}(V_1, V_2)) &\rightarrow V_1 & (\overline{\text{fst}}) \\ \overline{\text{snd}} (\overline{\text{pair}}(V_1, V_2)) &\rightarrow V_2 & (\overline{\text{snd}}) \end{aligned}$$

These correspond to the definitions in Haskell of function application and the standard functions for destructing pairs. So we do not have to add any Haskell to handle these.

3 Type-directed Partial Evaluation

The TDPE system itself is formulated as follows (Danvy 1996, Fig. 1); it can again be interpreted as a CRS:

$$\begin{aligned} \downarrow^b V &\rightarrow V & (\downarrow^b) \\ \downarrow^{\alpha \rightarrow \beta} V &\rightarrow \underline{\lambda x}. \downarrow^\beta (V^- (\uparrow_\alpha x)) & (\downarrow^\rightarrow) \\ \downarrow^{\alpha \times \beta} V &\rightarrow \underline{\text{pair}}(\downarrow^\alpha (\overline{\text{fst}} V), \downarrow^\beta (\overline{\text{snd}} V)) & (\downarrow^\times) \\ \uparrow^b E &\rightarrow E & (\uparrow^b) \\ \uparrow_{\alpha \rightarrow \beta} E &\rightarrow \overline{\lambda x}. \uparrow_\beta (E _ (\downarrow^\alpha V)) & (\uparrow^\rightarrow) \\ \uparrow_{\alpha \times \beta} E &\rightarrow \overline{\text{pair}}(\uparrow_\alpha (\underline{\text{fst}} E), \uparrow_\beta (\underline{\text{snd}} E)) & (\uparrow^\times) \end{aligned}$$

The two defined function symbols, \downarrow and \uparrow , are called *reification* and *reflection*, respectively.

The simplification proceeds by *type case analysis*. This is what Haskell offers in the form of *type class overloading*. Thus we can implement the system as a type class.

```
8 class ReifyReflect alpha where
9   reify  :: alpha -> Expr
10  reflect :: Expr -> alpha
```

3.1 Base Types

From the viewpoint of TDPE, all “uninteresting” types are base types in the sense that values and expressions are indistinguishable because the internal structure is not known. However, we cannot define a generic “base type instance.” (This is similar to the problem encountered in the Standard ML implementations of Zhe Yang and Andrzej Filinski.) Instead we have to pretend that all base values are

of type `Expr`. This means that we take the two rules at base type literally: the expression (or value) constructed by reification (or reflection, respectively) really *is* the base value (or expression, respectively), to convert. Thus the class instance looks as follows:

```
11 instance ReifyReflect Expr where
12   reify    v = v
13   reflect  e = e
```

Furthermore, Haskell overloading requires that we treat two cases separately: *type variables*, where we can ourselves decide the type, and *constants* where Haskell enforces a particular (primitive) type.

For type variables the problem is that Haskell does not permit free (native) type variables – the kind system is first order. Thus we will instead provide the user with pseudo type variables `Alpha`, `Beta`, ..., `Omega`. These are just aliased to the `Expr` type to make the reification be the identity on types as dictated by the definition.

```
14 type Alpha = Expr
15 type Beta  = Expr
   :
36 type Omega = Expr
```

Constants receive special treatment because we know how to convert them from values to expressions. It is an error to reflect a value of base type (we only handle “off-line” partial evaluation), unless this is an expression that already denotes a constant.

As an example constant type here is the definition for integers.

```
37 instance ReifyReflect Int where
38   reify v = BASE (show v)
39   reflect (BASE s) = read s
40   reflect _        = error "Cannot_reflect_non-constant_Int."
```

More are easily added, of course.

3.2 Function Types

Next the most fundamental type: function types. These are straightforward by following the definition and making the induction over types explicit:

```
41 instance (ReifyReflect alpha, ReifyReflect beta) =>
42   ReifyReflect (alpha → beta)
43 where
44   reify    v = LAMBDA (λx → reify    (apply v (reflect x)))
```

```
45 reflect e = lambda (λx → reflect (APPLY e (reify x)))
```

where the following dummy functions help emphasize the symmetry between data and code

```
46 lambda f = f
47 apply x y = x y
```

3.3 Product Types

Again the definition directly works, after the appropriate class constraints have been added.

```
48 instance (ReifyReflect alpha, ReifyReflect beta) ⇒
49           ReifyReflect (alpha,beta)
50 where
51   reify v = PAIR (reify (fst v)) (reify (snd v))
52   reflect e = pair (reflect (FST e)) (reflect (SND e))
```

where we have used the following auxiliary for symmetry.

```
53 pair x y = (x,y)
```

4 Partial Evaluation

Residualization is defined as follows in the original reference:

$$\text{residualize} = \text{statically-reduce} \circ \text{reify}$$

which, in our case, degenerates to the simple

```
54 residualize v = reify v
```

because all the static reductions are done internally by Haskell.

Other popular names for this function are also provided.

```
55 tdpe v = reify v
56 decompile v = reify v
```

Traditional partial evaluation, *i.e.*, specializing a program p to the static component s of a pair of inputs (s, d) , can now be expressed using currying:

```
57 pe p s = residualize (curry p s)
```

5 Some examples

Even this small language is interesting. In this session we reproduce the lines of a HUGS session (marked with small dots) to illustrate this.

- Hugs session for:
- /usr/local/share/hugs/lib/Prelude.hs
- TDPE.lhs

Let us look at the source of some standard Prelude.hs functions. Notice that we output the result as native Haskell so it can be reentered directly into HUGS (the actual function used to print the results is given in the appendix).

- TDPE> residualize (fst:: (Alpha,Beta) → Alpha)
- $\lambda a \rightarrow \text{fst } a$
- TDPE> residualize (snd:: (Alpha,Beta) → Beta)
- $\lambda a \rightarrow \text{snd } a$
- TDPE> residualize (curry:: ((Pi,Xi) → Mu) → (Pi → Xi → Mu))
- $\lambda a \ b \ c \rightarrow a \ (b,c)$
- TDPE> residualize (uncurry:: (Pi → Xi → Mu) → ((Pi,Xi) → Mu))
- $\lambda a \ b \rightarrow a \ (\text{fst } b) \ (\text{snd } b)$
- TDPE> residualize (id:: (Alpha → Alpha))
- $\lambda a \rightarrow a$
- TDPE> residualize (const:: (Alpha → Beta → Alpha))
- $\lambda a \ b \rightarrow a$
- TDPE> residualize ((.):: (Rho → Mu) → (Pi → Rho) → (Pi → Mu))
- $\lambda a \ b \ c \rightarrow a \ (b \ c)$
- TDPE> residualize (flip:: (Pi → Rho → Mu) → Rho → Pi → Mu)
- $\lambda a \ b \ c \rightarrow a \ c \ b$

We can even do “real” partial evaluation with this small system. Let us illustrate this with a generic example from Danvy (1996), the simple function bar that takes a static and a dynamic input:

```
58 bar (s,d) = d (5*s)
```

This we can specialize to a value for the static input by

- TDPE> pe (bar:: (Int,(Int → Alpha)) → Alpha) 100
- $\lambda a \rightarrow a \ 500$

And, finally, the canonical example: power, after suitable abstraction of the three numeric “primitives” used: one, sqr, and *:

```
59 power n one sqr (*) x = loop n where
60   loop n | n ≡ 0      = one
61          | odd n      = x * loop (n - 1)
62          | otherwise = sqr (loop (div n 2))
```

Partially evaluating this gives the expected result:

- TDPE> tdpe ((power 10):: Xi → (Xi → Xi) → (Xi → Xi → Xi) → Xi → Xi)
- $\lambda a \ b \ c \ d \rightarrow b \ (c \ d \ (b \ (b \ (c \ d \ a))))$

Or, more readably:

```
63 power_10 one sqr (*) x = sqr (x * (sqr (sqr (x * one))))
```

which we can use, of course, at different types.

```
· TDPE> power_10 1 (↑2) (*) 2
· 1024
· TDPE> power_10 1.0 (↑2) (*) 2.2
· 2655.99
```

6 Conclusions

We have shown how the simplest form of standard type-directed partial evaluation fits nicely with a pure functional language, and in particular how the type class mechanism of Haskell can be used to implement it elegantly: the entire source code of the system is included in this note!

Doing this, we also hope to have illustrated how easy it is to work with higher-order abstract syntax in functional languages.

We are currently extending and using this in several directions:

- We are experimenting with using the system for deforestation.
- More type forms should be included: in particular sum-types are challenging and require a significantly more complicated base CRS.
- It would be interesting to have an “on-line” version of the partial evaluator (Danvy 1998).

A Showing Terms

Nothing unusual is going on here: we merely include this to make the file self-contained and document how the terms above were printed. Really we should be using parsing and pretty-printing ...

```
64 instance Show Expr where
65   showsPrec _ e = st e vs0 where
```

We pick our variable names from an infinite supply.

```
66   vs0 = [ c:i | i ← ("":map show [1..]), c ← ['a'..'z'] ]
```

Our entry point prints a “top-level” expression.

```
67   st (LAMBDA f) (v:vs) = ss"\\".ss v.sb (f (BASE v)) vs
68   st e                vs = sa e vs
```

The next level takes care of outputting λ -terms as curried as possible.

```
69  sb (LAMBDA f) (v:vs)= ss"⌞".ss v.sb (f (BASE v)) vs
70  sb e          vs = ss"->".sa e vs
```

Applications associate to the left.

```
71  sa (APPLY e1 e2) vs = sa e1 vs.ss"⌞".si e2 vs
72  sa (FST e)      vs = ss"fst⌞".si e vs
73  sa (SND e)      vs = ss"snd⌞".si e vs
74  sa e            vs = si e vs
```

Finally, the innermost function ensures that anything complex is parenthesised.

```
75  si (PAIR e1 e2) vs = ss"(" .st e1 vs.ss"," .st e2 vs.ss")"
76  si (BASE s)     vs = ss s
77  si e            vs = ss"(" .st e vs.ss")"
```

All make use of the following abbreviation.

```
78  ss = showString
```

References

- Danvy, O. (1996), Type-directed partial evaluation, *in* G. L. Steele Jr., ed., ‘Proceedings of the Twenty-Third Annual ACM Symposium on Principles of Programming Languages’, ACM Press, St. Petersburg Beach, Florida, pp. 242–257.
- Danvy, O. (1998), Online type-directed partial evaluation, *in* M. Sato & Y. Toyama, eds, ‘Proceedings of the Third Fuji International Symposium on Functional and Logic Programming’, World Scientific, Kyoto, Japan. Extended version available as the technical report BRICS RS-97-53.
- Fasel, J. H., Hudak, P., Peyton Jones, S., Wadler, P. et al. (1992), ‘Haskell special issue’, *SIGPLAN Notices* **27**(5).
- Klop, J. W., van Oostrom, V. & van Raamsdonk, F. (1993), ‘Combinatory reduction systems: Introduction and survey’, *Theoretical Computer Science* **121**, 279–308.
- Nielson, F. & Nielson, H. R. (1992), *Two-Level Functional Languages*, Vol. 34 of *Cambridge Tracts in Theoretical Computer Science*, Cambridge University Press.